An Introduction to wxMaxima
(work in progress)

Peter K Nicol
peterknicol@gmail.com

November 17, 2025

Contents

Introduction

wxMaxima (wxM) is a front end for Maxima, a Computer Algebra System.
Other CASs are Mathematica, Maple, Matlab and Mathcad.
wxMaxima is an open source project that is free to use and works on MS Windows, Linux and Mac.
You can download a copy from

https://maxima.sourceforge.io/download.html

Windows: download maxima 5.48.1 (or later version)
or https://portableapps.com/node/23391 (portable application)

(the portable apps version will work on any computer running MS Windows but tends to be an older version).
When you open wxMaxima this is what you will see.



Close the message and open Edit Configure.





In MS Windows you should find that the box titled Maxima program is already completed but in Linux it won't be. (You will have to find the correct program (Maxima)).

You will find the program easier to use if you
(a) **untick** the box titled **Match parenthesis in text controls** and
(b) **tick** the box titled **Enter evaluates cells**.

Useful shortcuts	Zoom in **Alt-I**	Text box **CTRL-1** or **F6**
	Zoom out **Alt-O**	

Using wxMaxima

wxMaxima can do all the maths that you can do on paper. It is almost always correct[1] but will only give you the correct answer if it is given the correct question. This is what you have to learn to do. The main problem that you will have is to read the algebra correctly and then type it in so that the program reads what you want it to read.

The first input will be titled on the left by **%i1** where the **%** sign qualifies the **i** making it read as **i**nput **1**[2].On pressing Enter[3] a result will be displayed as **%o1** - **o**utput **1**. This means that Maxima has read the input and understood it. If the input has not been understood you will receive an error message (which may be almost impossible to understand). What you see on the output line should look very similar to what you would see printed or written - that is it should look correct.
For the documentation, please look at

https://maxima.sourceforge.io/documentation.html

The most comprehensive publication is the excellent **Maxima by Example** by Edwin L. (Ted) Woollett - see

http://www.csulb.edu/~woollett/

Download the pdf versions.

[1]Computer algebra systems are only as good as the people who wrote the program. Also they are prone to bugs - inadvertent errors usually resulting from changes made to improve the system.
[2]Note that %i1 is an input line but %i is $\sqrt{-1}$. %i1 is read as a single character
[3]or Shift-Enter if you haven't followed the instruction (b) above.

Let's try some simple maths.

We wish to calculate the value of $y = 3\sqrt{x}$ when $x = 7$. This can be done easily on a calculator but may take longer with wxM.



Well,this hasn't worked. You will have to use the same syntax as you would in a spreadsheet program.



All multiplication signs have to be typed in and every function name must be followed by brackets.



Right click and select Substitute.



The output will be $3\sqrt{7}$.

WxM will always attempt to give an exact answer, which this is. If you wish a decimal approximation you will need to do the following - right click **To Float**



The result will generally be rather a long number but watch out for exponentials[4].



You can go back and edit lines but the line number will then change as shown below



[4]for example: (%o13) 2.6602412030089932 10^{12} $\approx 2.660 \times 10^{12}$



Notice that line 3 has disappeared from the display. (Unfortunately) it's still there.

Notice the use of a text box.



A point worth remembering about.

Variables and Constants

As you will discover wxM can work with any variable but will default to x so be careful when calling up differentiation or integration. Also, in common with other CAS and usual practice, $T \neq t$.

```
(%i8) %e^(x*z)+1;
(%o8) %e^(x z) + 1
(%i9) subst(%i, x, (%o8) );
(%o9) %e^(%i z) + 1
(%i10) subst(%pi, z, (%o9) );
(%o10) 0
```

Constants π ($\approx 3.1459....$) and e ($\approx 2.71828.....$) will both be read as variables unless we tell wxM that they are constants. Hence, to input π type **%pi** and to input $3e^x$ input **3*%e^(x)** [5]. Also, j $(= \sqrt{-1})$ is typed as **%i**.

The interesting result $e^{i\pi} + 1 = 0$ (Euler's Identity)

[5] It is worth noting that in newer versions of wxM you will still have to type %e but it is possible to turn of the % sign in the output. This is maybe a bad idea.

Functions and Function Names

Most are obvious as they are exactly what you would use on a calculator, except that you have to type them in. Some are worthy of note though.

$\sin^{-1}$ on the calculator is typed as $a\sin$ - short for arcsine.

ln on the calculator is typed as log. This ($\log_e(x)$ or $\ln(x)$) is the only built in logarithmic function. $\log_{10}(x) = \dfrac{\log(x)}{\log(10)}$.

Please pay particular attention to the next part - function names are never followed by multiplication.

When inputting functions it must be remembered that function names are never followed by multiplication. Hence, If we want to differentiate $y = 5\sin(2x+1)$ we have to type as below





A correct result. Notice that this is the result $\dfrac{dy}{dx} = 10\cos(2x+1)$. There is no need for the left side of the equation.

If you do include the y it is, quite correctly[6] differentiated to 0. Unfortunately it doesn't make sense. Not a bug, but not very helpful.

[6] All CASs can do is partial differentiation



If your algebra is wrong this is the result you will get



Obviously %o1 is wrong, it doesn't look correct and you should know this.

Calculus

Differentiation

Very straightforward. Although it might seem best to select **Differentiate** from the menu, complete all the boxes and hit enter but it is not a good idea. If you have a syntax error you may not be alerted to this. Type in the function (RHS) and let wxM read the input first. If it is correct it will usually look correct. Check through the example above. Make sure that you are using the correct variable. Calculate the gradient of $y = 3x^2$ at $x = 7$





Input checked, Right click on **%o1**, select **Differentiate**. The variable is x and you want to differentiate once.

Note that Expression is given as the reference %o1 the output that you wish to differentiate.



Now, we wish to know the gradient at x = 7 - easy enough but can we get wxM to do it?



(%i1) 3*x^2;

(%o1) $3\,x^2$

(%i2) diff((%o1) ,x,1);

(%o2) $6\,x$

(%i3) subst(7, x, (%o2));

(%o3) 42

Old value is x and new value is 7. The answer is 42.[7]

[7]This value for x is only true locally, that is, it won't affect the use of x in lines below

Integration

Indefinite Integration

$\int 5x^3 + 4dx$









Notice that a constant of integration is not given. You will have to add it in by hand or go back and edit the input line (%o6 in this case)

Definite Integration

The procedure you adopt should add onto what you did above because the indefinite integral will be required for any handwritten work. Therefore if the question was $\int_1^3 5x^3 + 4dx$ it would be best to start with what we have above, to begin with, at least.



Notice: The Definite integration box is ticked.
The limits seem to be upside down but if you are reading the problem correctly then it works **from** left **to** right.
The Numerical integration box is not ticked
(similar to Simpson's Rule but more powerful).

(%i1) 5*x^3+4;
(%o1) $5\,x^3+4$
(%i2) integrate((%o1) , x);
(%o2) $\frac{5\,x^4}{4}+4\,x$
(%i3) integrate((%o1) , x, 1, 3);
(%o3) 108

(%i1) 5*x^3+4;
(%o1) $5\,x^3+4$
(%i2) integrate((%o1) , x, 1, 3);
(%o2) 108

I have integrated %o1 from $x = 1$ to $x = 3$. The total area is 108.
I could have gone straight from %o1 as on the right but I would have missed the integral which I might have needed to help with written work.

Differential Equations

The last question could have been phrased in a different way

Find y if $\dfrac{dy}{dx} = 5x^3 + 4$

(%i1) 'diff(y,x)=5*x^3+4;

(%o1) $\dfrac{d}{d\,x}y = 5\,x^3 + 4$

Notice: the apostrophe ' and it is the differential of y w.r.t. x.

Now, to solve it.





Careful here - the function is y and the variable is x.

(%i1) 'diff(y,x)=5*x^3+4;

(%o1) $\dfrac{d}{d\,x}y = 5\,x^3 + 4$

(%i2) ode2((%o1), y, x);

(%o2) $y = \dfrac{5\,x^4}{4} + 4\,x + \%c$

This time you have an additional term %c - a constant of integration.

If you have boundary conditions[8] you can proceed to obtain the particular equation.

[8] or observations



If $y = 20$ when $x = 1$ then
Notice that it states at the bottom of the screen

Solve initial value problem for first degree ODE





We now have an equation for y. It might require simplification if you are trying to write the steps down.





It is now possible to calculate the value for y from any value for x.

If $x = 3$ then (old value x, new value 3)



(%i1) 'diff(y,x)=5*x^3+4;

(%o1) $\frac{d}{d\,x}y = 5\,x^3 + 4$

(%i2) ode2((%o1), y, x);

(%o2) $y = \frac{5\,x^4}{4} + 4\,x + \%c$

(%i3) ic1((%o2), x=1, y=20);

(%o3) $y = \frac{5\,x^4 + 16\,x + 59}{4}$

(%i4) expand((%o3));

(%o4) $y = \frac{5\,x^4}{4} + 4\,x + \frac{59}{4}$

(%i5) subst(3, x, (%o4));

(%o5) $y = 128$

Plot Graph

You can only plot in 2D with 1 variable so copy and paste RHS[9]

(%i6) (5*x^4)/4+4*x+59/4;

(%o6) $\frac{5\,x^4}{4}+4\,x+\frac{59}{4}$

(%i7) wxplot2d([(%o6)], [x,-5,5])$

(%t7)



I have plotted an inline graph - easy to do but it is an image only. If you wish something (slightly) more useful then take off the wx.



A gnuplot has more options as shown but you will have to delete it before wxM will do any more work - it is recalculating every time the mouse is moved.
There will be more practice in this later.

[9]Right Hand Side of equals sign

Exercise 1

Use wxMaxima to...

1. Calculate the gradient of $y = 5e^{-0.2t}$ when $t = 3$.
2. Simplify
$$\frac{5x}{3y} + \frac{3}{2x}$$
to a single fraction and then expand - you should get back what you started with.

3. Plot, using wxplot2d, the function $y = 3\sin(2t)$ from $t = 0$ to $t = 1.5$ Calculate the area under $y = 3\sin(2t)$ between $t = 0$ to $t = 1.5$.
4. Solve the ODE
$$\frac{dy}{dt} = y\sin(2t)$$
with boundary conditions $t = 0$ when $y = 4$. Calculate the value of y when $t = 1.3$.

To solutions

Root finding

If you suspect that there may be positive and negative areas between the limits of a definite integration question you will want to check for roots

BUT first

plot the function between the roots to check if the are any roots (there may be more than one).

Example:

Calculate the area between the function $y = 5e^{(-0.5t)} \sin(2t)$ between $t = 3$ and $t = 4$.

```
(%i28) 5*%e^(-0.5*t)*sin(2*t);
(%o28) 5 %e^-0.5 t sin(2 t)
(%i30) wxplot2d([(%o28) ], [t,3,4])$
```

(%t30)



Although there is a Newton-Raphson procedure available it is usually quicker, and easier to get good solutions using the **Find Root** function (Right click and select from menu).

(%i28) 5*%e^(-0.5*t)*sin(2*t);

(%o28) $5\,\%e^{-0.5\,t}\,\sin(2\,t)$

(%i30) wxplot2d([(%o28)], [t,3,4])$

(%t30) 

root between t = 3 and t = 4

(%i31) find_root((%o28) , t, 3, 4);

(%o31) 3.141592653589793

which is, of course, the correct answer

(%i32) %pi;

(%o32) π

(%i33) float((%o32)), numer;

(%o33) 3.141592653589793

(%i35) newton(5*%e^(-0.5*t)*sin(2*t),t,3,0.0000000001);

(%o35) 3.141592653589793

Commentary Note : You must load the package newton1 before using the commande `newton`. This is done with `load(newton1)`.

Solving Systems of Equations

Whether they are simultaneous linear equations or a mixture of types they can be solved using the **Solve** function. This is one of the rare cases where you have to call up the function first before typing in the expressions



Solve $\begin{cases} 5x + 3y = 7 \\ 2x - 5y = 4 \end{cases}$

Notice the comma between the equations



Notice the comma between the variables.

The result - (maybe float not required)

(%i1) solve([5*x+3*y=7,2*x-5*y=4], [x,y]);

(%o1) $[[x=\frac{47}{31}, y=-\frac{6}{31}]]$

(%i2) float(%o1), numer;

(%o2) $[[x=1.516129032258065, y=-0.19354838709677]]$

Matrices

To solve the previous example using matrices

$$\begin{bmatrix} 5 & 3 \\ 2 & -5 \end{bmatrix} \begin{bmatrix} x \\ y \end{bmatrix} = \begin{bmatrix} 7 \\ 4 \end{bmatrix}$$









The complete solution

Algebra, enter matrix, 2 by 2

```
(%i3) matrix(
       [5,3],
       [2,-5]
      );
```

$$(\%o3)\ \begin{bmatrix} 5 & 3 \\ 2 & -5 \end{bmatrix}$$

Algebra, invert matrix

```
(%i4) invert((%o3));
```

$$(\%o4)\ \begin{bmatrix} \frac{5}{31} & \frac{3}{31} \\ \frac{2}{31} & -\frac{5}{31} \end{bmatrix}$$

Algebra, enter 2 by 1 matrix

```
(%i5) matrix(
       [7],
       [4]
      );
```

$$(\%o5)\ \begin{bmatrix} 7 \\ 4 \end{bmatrix}$$

multiply inverse on left by RHS on right

Use . to multiply matrices, not *

```
(%i6) %o4.%o5;
```

$$(\%o6)\ \begin{bmatrix} \frac{47}{31} \\ -\frac{6}{31} \end{bmatrix}$$

```
(%i7) float((%o6) ), numer;
```

$$(\%o7)\ \begin{bmatrix} 1.516129032258065 \\ -0.19354838709677 \end{bmatrix}$$

x = 1.516, y = -0.1935 (4 sf)

Note the use of **.** to multiply matrices[10]!

You can easily achieve a spurious result!

[10]Matrix multiplication is non-commutative.

(%i14) matrix(
[2,3],
[4,5]
);

(%o14) $\begin{bmatrix} 2 & 3 \\ 4 & 5 \end{bmatrix}$

(%i15) invert((%o14));

(%o15) $\begin{bmatrix} -\frac{5}{2} & \frac{3}{2} \\ 2 & -1 \end{bmatrix}$

(%i16) %o14*%o15;

(%o16) $\begin{bmatrix} -5 & \frac{9}{2} \\ 8 & -5 \end{bmatrix}$

Wrong answer but it has done what was asked and multiplied corresponding entries

(%i17) %o14.%o15;

(%o17) $\begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}$

correct answer, a matrix times its inverse = the identity matrix

Exercise 2

1. Find the area between the r axis and the curve $w = 3e^{(0.5r)} - 7$ between $r = 0$ and $r = 3$.
2. Solve the system of equations

$$\begin{cases} 3x + 4y = 10 \\ x^2 - y = 3 \end{cases}$$

3. Solve for x, y and z, using matrices.

$$\begin{cases} 3x - 2y + 7z = 27 \\ 5x - y - 3z = -10 \\ -x + 3y - z = 2 \end{cases}$$

To solutions

Ordinary Differential Equations

The ODE solver in wxM is able to handle 1st and 2nd order ODEs with boundary conditions.

For instance $\dfrac{dy}{dt} - 5y = 3$ boundary conditions $t = 0,\ y = 7$



Notice the apostrophe before diff, also y then t..





We now have the general equation





The whole solution

(%i1) 'diff(y,t)-5*y=3;

(%o1) $\frac{d}{dt}y-5y=3$

(%i2) ode2(%o1), y, t);

(%o2) $y=\left(\%c-\frac{3\,\%e^{-5t}}{5}\right)\%e^{5t}$

(%i3) ic1(%o2), t=0, y=7);

(%o3) $y=\frac{38\,\%e^{5t}-3}{5}$

(%i4) expand(%o3));

(%o4) $y=\frac{38\,\%e^{5t}}{5}-\frac{3}{5}$

There are, of course, other methods of solving this by hand which may be considered to be quicker and easier.

$$2\frac{d^2y}{dt^2}-3\frac{dy}{dt}+4y=2\sin(t) \text{ when } t=0, y=3 \text{ and } \frac{dy}{dt}=5$$

Solve for y and calculate the value of y when $t=0.3$.



Equations - Initial Value Problem (2) (not 1)

Notice the change from x to t.

(%i5) 2*'diff(y,t,2)-3*'diff(y,t)+4*y=2*sin(t);

(%o5) $2\left(\frac{d^2}{d t^2} y\right) - 3\left(\frac{d}{d t} y\right) + 4 y = 2 \sin(t)$

(%i6) ode2((%o5), y, t);

(%o6) $y = \%e^{\frac{3t}{4}}\left(\%k1 \sin\left(\frac{\sqrt{23}\, t}{4}\right) + \%k2 \cos\left(\frac{\sqrt{23}\, t}{4}\right)\right) + \frac{4 \sin(t) + 6 \cos(t)}{13}$

(%i7) ic2((%o6), t=0, y=3, 'diff(y,t)=5);

(%o7) $y = \%e^{\frac{3t}{4}}\left(\frac{145 \sin\left(\frac{\sqrt{23}\, t}{4}\right)}{13\sqrt{23}} + \frac{33 \cos\left(\frac{\sqrt{23}\, t}{4}\right)}{13}\right) + \frac{4 \sin(t) + 6 \cos(t)}{13}$

(%i8) subst(0.3, t, (%o7));

(%o8) $y = 1.252322716191864\left(\frac{145 \sin(0.075\sqrt{23})}{13\sqrt{23}} + \frac{33 \cos(0.075\sqrt{23})}{13}\right) + 0.53185382779992$

(%i9) float((%o8)), numer;

(%o9) $y = 4.532567143338392$

Sub in $t = 0.3$ and float.

If you wish to produce a graph of the result you will only require the RHS.

(%i10) %e^((3*t)/4)*((145*sin((sqrt(23)*t)/4))/(13*sqrt(23))+(33*cos((sqrt(23)*t)/4))/13)+(4*sin(t)+6*cos(t))/13;

(%o10) $\%e^{\frac{3t}{4}}\left(\frac{145 \sin\left(\frac{\sqrt{23}\, t}{4}\right)}{13\sqrt{23}} + \frac{33 \cos\left(\frac{\sqrt{23}\, t}{4}\right)}{13}\right) + \frac{4 \sin(t) + 6 \cos(t)}{13}$

(%i11) wxplot2d([(%o10)], [t,-5,5])$

(%t11)



You can change the limits to see more or less. Below I have inserted values for y to clip the vertical range.

(%i12) wxplot2d([(%o10)], [t,-5,15], [y,-60,60])$

plot2d: some values were clipped.

(%t12) 

Laplace Transforms

Having worked out, by hand, $L[y]$, wxM can then be used to find y.

Also it is easy to find $L[y]$ as shown below.

Taking the result of the last example :

(%i20) %e^((3*t)/4)*((145*sin((sqrt(23)*t)/4))/(13*sqrt(23))+(33*cos((sqrt(23)*t)/4))/13)+(4*sin(t)+6*cos(t))/13;

$$(\%o20)\ \%e^{\frac{3t}{4}}\left(\frac{145\sin\left(\frac{\sqrt{23}\,t}{4}\right)}{13\sqrt{23}}+\frac{33\cos\left(\frac{\sqrt{23}\,t}{4}\right)}{13}\right)+\frac{4\sin(t)+6\cos(t)}{13}$$

(%i21) laplace((%o20), t, s);

$$(\%o21)\ \frac{66s+23}{26s^2-39s+52}+\frac{\frac{6s}{s^2+1}+\frac{4}{s^2+1}}{13}$$

(%i22) ratsimp((%o21));

$$(\%o22)\ \frac{6s^3+s^2+6s+3}{2s^4-3s^3+6s^2-3s+4}$$

(%i28) partfrac((%o21), s);

$$(\%o28)\ \frac{66s+23}{13\,(2s^2-3s+4)}+\frac{6s+4}{13\,(s^2+1)}$$

Note, also, the use of partial fractions.

So, starting at the beginning

Example : $\dfrac{d^2y}{dt^2}-5\dfrac{dy}{dt}+6y=4 \qquad [t=0, y=0]$

It should be relatively easy to obtain

$$L[y]=\frac{4}{s(s^2-5s+6)}$$

and the next stage is no harder:



I have typed in the RHS of $L[y]$ and the first thing to try is decomposing it with partial fractions as shown. Of course, it is easy to obtain y directly but if you wish to understand the structure of the solution you need to go further.

(%i31) 4/(s*(s^2-5*s+6));

(%o31) $\frac{4}{s\,(s^2-5\,s+6)}$

(%i32) partfrac((%o31), s);

(%o32) $\frac{2}{3\,s}-\frac{2}{s-2}+\frac{4}{3\,(s-3)}$

Now,it would be easy to obtain y using the standard table of transforms but wxM can do it for us either as a solution or as a check on a handwritten solution.





(%i31) 4/(s*(s^2-5*s+6));

(%o31) $\frac{4}{s\,(s^2-5\,s+6)}$

(%i32) partfrac((%o31), s);

(%o32) $\frac{2}{3\,s}-\frac{2}{s-2}+\frac{4}{3\,(s-3)}$

(%i33) ilt((%o32), s, t);

(%o33) $\frac{4\,\%e^{3\,t}}{3}-2\,\%e^{2\,t}+\frac{2}{3}$

The complete solution is shown above.

It is easy to go backwards as well.

(%i33) ilt((%o32), s, t);

(%o33) $\frac{4\,\%e^{3\,t}}{3}-2\,\%e^{2\,t}+\frac{2}{3}$

(%i34) laplace((%o33), t, s);

(%o34) $\frac{2}{3\,s}-\frac{2}{s-2}+\frac{4}{3\,(s-3)}$

(%i35) ratsimp((%o34));

(%o35) $\frac{4}{s^3-5\,s^2+6\,s}$

A useful way to check your work.

Solutions

Exercise 1

Back to Exercise 1

1.
```
(%i1) 5*%e^(-0.2*t);
```
(%o1) $5\,\%e^{-0.2\,t}$
```
(%i2) diff((%o1) ,t,1);
```
(%o2) $-1.0\,\%e^{-0.2\,t}$
```
(%i3) subst(3, t, (%o2) );
```
(%o3) -0.54881163609403

   Did you remember to place a % in front of e and swap x for t?

2.
```
(%i4) 5*x/(3*y)+3/(2*x);
```
(%o4) $\frac{5\,x}{3\,y}+\frac{3}{2\,x}$
```
(%i5) ratsimp((%o4) );
```
(%o5) $\frac{9\,y+10\,x^2}{6\,x\,y}$
```
(%i6) expand((%o5) );
```
(%o6) $\frac{5\,x}{3\,y}+\frac{3}{2\,x}$

   I have typed this in with the minimum number of brackets.

3.

(%i5) 3*sin(2*t);

(%o5) 3 sin(2 t)

Always plot between limits - check for roots - positive and negative areas
Change variable to t.

(%i8) wxplot2d([(%o5)], [t,-0,1.5])$

(%t8)



no roots - one area - go ahead
Change variable to t

(%i9) integrate((%o5) , t, 0, 1.5);

rat: replaced 1.5 by 3/2 = 1.5
rat: replaced 1.5 by 3/2 = 1.5
rat: replaced 1.5 by 3/2 = 1.5
rat: replaced 1.5 by 3/2 = 1.5
rat: replaced 1.5 by 3/2 = 1.5
rat: replaced 1.5 by 3/2 = 1.5
rat: replaced 1.5 by 3/2 = 1.5
rat: replaced 1.5 by 3/2 = 1.5
rat: replaced 3.0 by 3/1 = 3.0
rat: replaced -3.0 by -3/1 = -3.0
rat: replaced 3.0 by 3/1 = 3.0
rat: replaced 0.49499624830022 by 3957/7994 = 0.49499624718539

(%o9) $\frac{11931}{3997}$

exact answer given - float

(%i10) float((%o9)), numer;

(%o10) 2.984988741556167

4.

(%i11) 'diff(y,t)=y*sin(2*t);

(%o11) $\frac{d}{d\,t}y = \sin(2\,t)\,y$

(%i12) ode2((%o11), y, t);

(%o12) $y = \%c\ \%e^{-\frac{\cos(2\,t)}{2}}$

(%i13) ic1((%o12), t=0, y=4);

(%o13) $y = 4\ \%e^{1/2-\frac{\cos(2\,t)}{2}}$

(%i14) subst(1.3, t, (%o13));

(%o14) $y = 10.12227800467399$

Exercise 2

Back to Exercise 2

1.

(%i13) 3*%e^(0.5*t)-7;

(%o13) $3\,\%e^{0.5\,t}-7$

Plot between limits, Roots?

(%i14) wxplot2d([(%o13)], [t,0,3])$

(%t14)



Find root

(%i15) find_root((%o13) , t, 0, 3);

(%o15) 1.694595720774407

integrate between left limit and root

(%i16) integrate((%o13) , t, 0, 1.694595720774407);

rat: replaced 1.694595720774407 by 4045/2387 = 1.694595726853792
rat: replaced 1.694595720774407 by 4045/2387 = 1.694595726853792
rat: replaced 1.694595720774407 by 4045/2387 = 1.694595726853792
rat: replaced 0.5 by 1/2 = 0.5
rat: replaced 0.5 by 1/2 = 0.5
rat: replaced -0.5 by -1/2 = -0.5
rat: replaced 0.5 by 1/2 = 0.5
rat: replaced 6.0 by 6/1 = 6.0
rat: replaced 2.13782995457915 by 729/341 = 2.13782991202346

(%o16) $-\frac{1317}{341}$

integrate between root and right limit

(%i17) integrate((%o13) , t, 1.694595720774407, 3);

rat: replaced 1.305404279225593 by 3116/2387 = 1.305404273146209
rat: replaced 1.694595720774407 by 4045/2387 = 1.694595726853792
rat: replaced 1.305404279225593 by 3116/2387 = 1.305404273146209
rat: replaced 0.5 by 1/2 = 0.5
rat: replaced 0.5 by 1/2 = 0.5
rat: replaced -0.5 by -1/2 = -0.5
rat: replaced 0.5 by 1/2 = 0.5
rat: replaced 2.13782995457915 by 729/341 = 2.13782991202346
rat: replaced 5.890134422028389 by 2627/446 = 5.890134529147982

(%o17) $\frac{570673}{152086}$

Add moduli

(%i18) %o17-%o16;

(%o18) $\frac{1158055}{152086}$

(%i19) float((%o18)), numer;

(%o19) 7.614474705101062

2.

(%i7) solve([3*x+4*y=10,x^2-y=3], [x,y]);

(%o7) $[[x=2, y=1], [x=-\frac{11}{4}, y=\frac{73}{16}]]$

(%i8) float((%o7)), numer;

(%o8) $[[x=2.0, y=1.0], [x=-2.75, y=4.5625]]$

(%i2) 3*x+4*y=10;

(%o2) $4y+3x=10$

(%i3) solve([(%o2)], [y]);

(%o3) $[y=-\frac{3x-10}{4}]$

(%i4) x^2-y=3;

(%o4) $x^2-y=3$

(%i5) solve([(%o4)], [y]);

(%o5) $[y=x^2-3]$

(%i6) wxplot2d([-(3*x-10)/4,x^2-3], [x,-5,5])$

(%t6)



3.

```
(%i8) matrix(
      [3,-2,7],
      [5,-1,-3],
      [-1,3,-1]
      );
```

$$(\%o8)\quad \begin{bmatrix} 3 & -2 & 7 \\ 5 & -1 & -3 \\ -1 & 3 & -1 \end{bmatrix}$$

```
(%i9) invert((%o8));
```

$$(\%o9)\quad \begin{bmatrix} \frac{5}{56} & \frac{19}{112} & \frac{13}{112} \\ \frac{1}{14} & \frac{1}{28} & \frac{11}{28} \\ \frac{1}{8} & -\frac{1}{16} & \frac{1}{16} \end{bmatrix}$$

```
(%i10) matrix(
       [27],
       [-10],
       [2]
       );
```

$$(\%o10)\quad \begin{bmatrix} 27 \\ -10 \\ 2 \end{bmatrix}$$

```
(%i11) %o9.%o10;
```

$$(\%o11)\quad \begin{bmatrix} \frac{53}{56} \\ \frac{33}{14} \\ \frac{33}{8} \end{bmatrix}$$

```
(%i12) float((%o11) ), numer;
```

$$(\%o12)\quad \begin{bmatrix} 0.94642857142857 \\ 2.357142857142857 \\ 4.125 \end{bmatrix}$$

x = 0.9464, y = 2.357, z = 2.125 (4sf)3.

$$x = 0.9464, y = 2.357, z = 4.125$$

Latest News for wxMaxima

The latest version of wxMaxima is the 25.04 (November 2025). This version is packaged with the latest version of Maxima (the 5.48.1). The site for wxMaxima :

https://wxmaxima-developers.github.io/wxmaxima/index.html

This version offers tool palettes, many new menu entries, the ability to insert images into the spreadsheet, as well as the capability to structure a document with headings, sections in addition to the text.